Iterációk, iterált pontsorozatok

1D

n-edik iterált
Legyen f, x rögzített.

x nulladik iterált: $f^{(0)}[x]$

1. iterált $f^{(1)}[x] = f[x]$
2. iterált $f^{(2)}[x] = f[f[x]]$

Mathematica függvények: Nest[f,x0,n] -- $f^{(n)}$[x0]

```
In[1]:= Nest[f, x, 3]
Out[1]= f[f[f[x]]]

In[2]:= NestList[f, x, 3]
Out[2]= {x, f[x], f[f[x]], f[f[f[x]]]}

In[3]:= g[x_] := x^2 + 1;

In[4]:= NestList[g, 0, 4]
Out[4]= {0, 1, 2, 5, 26}

In[5]:= Nest[g, 0, 4]
Out[5]= 26
```

Orbitok

speciális orbitok, ciklusok
A vizsgált nemlineáris leképezés-családot logisztikus leképezésnek (Logistic Map) nevezzük.(r -valós paraméter)

```
In[6]:= LM[r_, x_] := r x (1 - x);

In[7]:= LM[4, x]
Out[7]= 4 (1 - x) x

In[8]:= Expand[%]
```

Out[8]= $4\,x - 4\,x^2$

In[9]:=
```
NestList[LM[4, #] &, 3 / 4, 5]
```

Out[9]= $\left\{\frac{3}{4}, \frac{3}{4}, \frac{3}{4}, \frac{3}{4}, \frac{3}{4}, \frac{3}{4}\right\}$

Észrevétel: A iterált sorozat minden tagja ugyanaz $\Longrightarrow$
Def. $x_0$az f függvény fixpontja, ha $f[x_0] = x_0$

In[10]:=
```
Solve[LM[4, x] == x, x]
```

Out[10]= $\left\{\{x \to 0\}, \left\{x \to \frac{3}{4}\right\}\right\}$

A 4 paraméterû logisztikus leképezésnek 2 fixpontja van: 0, 3/4

In[11]:=
```
NestList[LM[4, #] &, 0, 5]
```

Out[11]= {0, 0, 0, 0, 0, 0}

Egy kicsit érdekesebb iterált sorozat: vegyük észre,, hogy a kezdôérték nem racionális!

In[15]:=
```
NestList[LM[4, #] &, 1 / 8 (5 - Sqrt[5]), 6] // Simplify // TableForm
```

Out[15]//TableForm=

$\frac{1}{8}\left(5-\sqrt{5}\right)$

$\frac{1}{8}\left(5+\sqrt{5}\right)$

$\frac{1}{8}\left(5-\sqrt{5}\right)$

$\frac{1}{8}\left(5+\sqrt{5}\right)$

$\frac{1}{8}\left(5-\sqrt{5}\right)$

$\frac{1}{8}\left(5+\sqrt{5}\right)$

$\frac{1}{8}\left(5-\sqrt{5}\right)$

In[16]:=
```
NestList[LM[4, LM[4, #]] &, 1 / 8 (5 - Sqrt[5]), 6] // Simplify // TableForm
```

Out[16]//TableForm=

$\frac{1}{8}\left(5-\sqrt{5}\right)$

$\frac{1}{8}\left(5-\sqrt{5}\right)$

$\frac{1}{8}\left(5-\sqrt{5}\right)$

$\frac{1}{8}\left(5-\sqrt{5}\right)$

$\frac{1}{8}\left(5-\sqrt{5}\right)$

$\frac{1}{8}\left(5-\sqrt{5}\right)$

$\frac{1}{8}\left(5-\sqrt{5}\right)$

Észrevétel: ismétlôdés, 2-ciklus. x0 m'sodik iteráltjának fixpontja!

Mi történik, ha más x_0 kezdôértéket választunk?

Szabályos/kaotikus viselkedés konvergencia?

```
In[17]:= NestList[LM[2, #] &, 1/2, 4]
```

Out[17]= $\left\{\frac{1}{2}, \frac{1}{2}, \frac{1}{2}, \frac{1}{2}, \frac{1}{2}\right\}$

```
In[18]:= N[%]

Out[18]= {0.5, 0.5, 0.5, 0.5, 0.5}

In[19]:= NestList[LM[2, #] &, 1/4, 10] // N

Out[19]= {0.25, 0.375, 0.46875, 0.498047, 0.499992, 0.5, 0.5, 0.5, 0.5, 0.5, 0.5}

In[20]:= NestList[LM[2, #] &, 1/10, 10] // N

Out[20]= {0.1, 0.18, 0.2952, 0.416114, 0.485926, 0.499604, 0.5, 0.5, 0.5, 0.5, 0.5}
```

Észrevétel: nincs ismétlôdés, de konvergencia figyelhetô meg, a limesz fixpont.

```
In[21]:= NestList[LM[4, #] &, 3/4, 4]
```

Out[21]= $\left\{\frac{3}{4}, \frac{3}{4}, \frac{3}{4}, \frac{3}{4}, \frac{3}{4}\right\}$

```
In[22]:= N[%]

Out[22]= {0.75, 0.75, 0.75, 0.75, 0.75}

In[23]:= NestList[LM[4, #] &, 7/10, 10];

In[24]:= N[%]

Out[24]= {0.7, 0.84, 0.5376, 0.994345, 0.0224922,
          0.0879454, 0.320844, 0.871612, 0.447617, 0.989024, 0.0434219}

In[27]:= NestList[LM[4, #] &, 74/100, 10];

In[26]:= N[%]

Out[26]= {0.74, 0.7696, 0.709263, 0.824835, 0.577928,
          0.975709, 0.0948039, 0.343264, 0.901736, 0.354433, 0.915241}
```

Észrevétel: nincs ismétlôdés, konvergencia

2D

Ugyanez vektorokra $F : \mathbb{R}^2 \to \mathbb{R}^2$

```
In[28]:= F[{x_, y_}] := {x + 1, 2 y}
```

In[29]:=
```
NestList[F, {0, 1}, 5] // TableForm
```

Out[29]//TableForm=

0	1
1	2
2	4
3	8
4	16
5	32

Kitekintés

Hasonló iterációs eljárások (összefoglaló):

- $R^* R$ iteráció sajátérték közelítésére
- Newton-Raphson gyökök (egyenletek, egyenletrendszerek megoldásának) közelítésére
- Kaczmarz-Steinhaus lineáris egyenletrendszer megoldásának közelítésére
- Jacobi, Gauss-Seidel lineáris egyenletrendszer megoldásának közelítésére

---

Newton-Raphson

1D

Adott f[x]=0 egy. továbbá egy x^* valós gyök "durva" közelítése: x0

Pl. $x^2 - 2 = 0$ x0=1

F[x]=x-f[x] / f ' [x]

Áll. Bizonyos feltételek mellett $F^{(n)}[x0] \longrightarrow x^*$

Példa.

In[30]:=
```
NewtonRaphsonStep[x_] := x - f[x] / f'[x];
```

In[31]:=
```
f[x_] := x^2 - 2;
x0 = 2;
```

In[33]:=
```
NestList[NewtonRaphsonStep, x0, 6]
```

Out[33]= $\left\{2, \frac{3}{2}, \frac{17}{12}, \frac{577}{408}, \frac{665\,857}{470\,832}, \frac{886\,731\,088\,897}{627\,013\,566\,048}, \frac{1\,572\,584\,048\,032\,918\,633\,353\,217}{1\,111\,984\,844\,349\,868\,137\,938\,112}\right\}$

In[34]:=
```
N[%]
```

Out[34]=
```
{2., 1.5, 1.41667, 1.41422, 1.41421, 1.41421, 1.41421}
```

In[35]:= N[$\sqrt{2}$]

Out[35]= 1.41421

In[36]:= Clear[x]

In[37]:= NewtonRaphsonStep[x] // Simplify

Out[37]= $\frac{1}{x} + \frac{x}{2}$

Megjegyzés. Ez ugyanaz, mint az elôadáson megismert $\frac{1}{2}\left(x + \frac{2}{x}\right)$ képlet

In[38]:= NestList[NewtonRaphsonStep, 17 / 12, 10] // N

Out[38]= {1.41667, 1.41422, 1.41421, 1.41421, 1.41421, 1.41421, 1.41421, 1.41421, 1.41421, 1.41421, 1.41421}

Empirikus konvergencia-sebesség vizsgálat

In[39]:= Transpose[{N[NestList[NewtonRaphsonStep, x0, 6], 24], Table[N[$\sqrt{2}$, 24], {7}]}] // TableForm

Out[39]//TableForm=

2.00000000000000000000000	1.41421356237309504880169
1.50000000000000000000000	1.41421356237309504880169
1.41666666666666666666667	1.41421356237309504880169
1.41421568627450980392157	1.41421356237309504880169
1.41421356237468991062630	1.41421356237309504880169
1.41421356237309504880169	1.41421356237309504880169
1.41421356237309504880169	1.41421356237309504880169

Figyeljük meg, hogy hány hogyan változik a pontos jegyek száma
Fourier monton-konvergencia tételei, kvadratikus konvergencia seb.

In[40]:= Resolve[ForAll[x, x > Sqrt[2], 1 / x + x / 2 > Sqrt[2]], Reals]

Out[40]= True

In[41]:= Resolve[ForAll[x, 0 < x < Sqrt[2], 1 / x + x / 2 > Sqrt[2]], Reals]

Out[41]= True

In[42]:= Resolve[ForAll[x, x > Sqrt[2], 1 / x + x / 2 - Sqrt[2] < x - Sqrt[2]], Reals]

Out[42]= True

- **Feladat: $\sqrt[3]{7}$ közelítése**

In[43]:= Clear[f];

In[44]:= f[x_] := x^3 - 7

In[48]:= NestList[NewtonRaphsonStep, 3, 5] // N

Out[48]= {3., 2.25926, 1.96331, 1.91421, 1.91293, 1.91293}

In[49]:= `N[7^(1/3)]`

Out[49]= `1.91293`

- **Vessük össze!**

In[50]:= `Solve[6 - x^2 == 3 x^2 - 3]`

Out[50]= $\left\{\left\{x \to -\frac{3}{2}\right\}, \left\{x \to \frac{3}{2}\right\}\right\}$

In[52]:= `Solve[Cos[x] == x]`

Solve::tdep : The equations appear to involve the variables to be solved for in an essentially non−algebraic way. ≫

Out[52]= `Solve[Cos[x] == x]`

Oldjuk meg numerikusan, közelítsük a zérushely(ek)et iterációval, készítsünk ábrát a kezdõérték megállapítására!
Használhatjuk az ábrát egy kezdõérték abszcisszájának leolvasására is!

In[54]:= `Plot[{Cos[x], x}, {x, -5, 5}, PlotRange → {-5, 5}]`

Out[54]=



Legyen f kétszer diff. ∃x∈[a,b]: f[x]=0? pl folytonossági megfontolások
Alapötlet: legyen $x_0$egy jó közelítése a gyöknek és használjuk a deriváltat egy jobb közelítés megadására

$$f'[x_n] = (0 - f[x_n])/(x_{n+1} - x_n) \Longrightarrow x_{n+1} = x_n + \frac{-f[x_n]}{f'[x_n]}$$

Egy lépés az iterációban (f globális)

In[55]:=
```
x0 = 1;
f[x_] := Cos[x] - x;
```

In[57]:=
```
NewtonRaphsonStep[x_] := N[x - f[x] / f'[x]];
```

In[58]:= `Nest[NewtonRaphsonStep, x0, 2]`

Out[58]= `0.739113`

In[59]:= `FindRoot[f[x] == 0, {x, x0}]`

Out[59]= `{x → 0.739085}`

In[60]:=
```
NestList[NewtonRaphsonStep, x0, 3]
```
Out[60]=
```
{1, 0.750364, 0.739113, 0.739085}
```

Ezt egy kicsit nehezebb megérteni, mindenki próbálkozzon azért!

In[61]:=
```
NestWhile[NewtonRaphsonStep, x0, Unequal, 2]
```
Out[61]=
```
0.739085
```

Mathematica függvény

In[62]:=
```
FindRoot[f[x] == 0, {x, x0}]
```
Out[62]=
```
{x → 0.739085}
```

▪ Feladat

Adjuk meg 3 tizedesjegy pontossággal, felhasználva a NewtonRaphsonStep fgv-t, az alábbi egyenletek összes zéróhelyét, melyek benne vanak a 0 3 sugarú környezetében.

$$eq1 = x^2 == 5;$$
$$eq2 = x^3 == 2;$$
$$eq3 = 4\,x\,\mathrm{Cos}[x]^2 + \pi/2 == 2\,x + \mathrm{Sin}[2\,x];$$

In[63]:=
$$f[x\_] := 4\,x\,\mathrm{Cos}[x]^2 + \pi/2 - (2\,x + \mathrm{Sin}[2\,x]);$$

In[64]:=
```
Plot[f[x], {x, -3, 3}]
```
Out[64]=


In[65]:=
```
{Nest[NewtonRaphsonStep, -2, 3], Nest[NewtonRaphsonStep, 1, 3], Nest[NewtonRaphsonStep, 2, 3]}
```
Out[65]=
```
{-2.41614, 0.952848, 2.04516}
```

▪ Projektmunka: NR Vizualizáció

Egy lépés viz.

In[153]:=
```
Clear[f, g]
```
In[154]:=
```
f[x_] := x^2 - 5;
```

```
g[x_] := x^3 - 2;

D[f[y], y]
```

In[157]:=

```
f[x]
```

Out[157]= $-5+x^2$

In[165]:= `NewtonRaphsonStep[x]`

Out[165]= $x - \frac{0.5\left(-5.+x^2\right)}{x}$

függvény grafikonja, 2 szakasz, 1 pont

In[177]:=
```
NewtonRaphsonVis[f_, x0_] := Module[{x}, Plot[f[x], {x, x0 - 1, x0 + 1}, PlotStyle → {Green},
   Epilog → {PointSize[.03], Red, Point[{x0, 0}], Blue, Point[{NewtonRaphsonStep[x0], 0}]}]]
```

In[178]:= `NewtonRaphsonVis[f, 3]`

Out[178]=




```
NewtonRaphsonVis[f, 3];
```



```
Nest[NewtonRaphsonStep, 3, 1]
```

```
2.33333
```

f globális!

```
Clear[f]; f = g;
```

```
Nest[NewtonRaphsonStep, 2, 1]
```

```
1.5
```

Több lépés

```
f[x]
```

```
NewtonRaphsonVis[f_, x0_, step_] := Module[
  {loctable = Transpose[NestList[NewtonRaphsonStep, x0, step] /. x_ -> {x, f[x]}], loctable2},
  loctable2 = Drop[loctable, 1] /. {x_?NumericQ, y_} → {x, 0};
  Plot[f[x], {x, x0 - 5, x0 + 1}, PlotStyle → {Thickness[.01], RGBColor[0, 1, 0]},
   AspectRatio → .75, Epilog → {RGBColor[1, 0, 0], PointSize[.02],
     Drop[loctable, -1] /. {x_?NumericQ, y_?NumericQ} → Point[{x, y}],
     RGBColor[0, 0, 1], loctable2 /. {x_?NumericQ, 0} → Point[{x, 0}],
     Map[Line[#] &, Transpose[{Drop[loctable, -1], loctable2}]],
     Map[Line[#] &, Transpose[{Drop[loctable, 1], loctable2}]]}, ImageSize → {600, 600}]]
```

```
NewtonRaphsonVis[f, 7, 3]
```




```
NestList[NewtonRaphsonStep, 7, 3] // N
```

■ ℂ: Kitekintés: Newton fraktálok

In[83]:=
```
NoOfIterations = 6; GridSize = .1;
```

In[84]:=
```
F[z_] := z^3 - 2;
```

In[85]:=
```
NS[z_] := z - F[z] / F'[z]
```

In[86]:= `NestList[NS, 1 + I, 10] // N // Chop // TableForm`

Out[86]//TableForm=

```
1. + 1. i
0.666667 + 0.333333 i
1.16444 - 0.737778 i
0.92614 - 0.17463 i
1.31644 + 0.156907 i
1.2463 + 0.0154548 i
1.25987 - 0.000338272 i
1.25992 + 2.60235 × 10^-8 i
1.25992
1.25992
1.25992
```

In[87]:= `NestList[NS, I, 10] // N // Chop // TableForm`

Out[87]//TableForm=

```
1. i
-0.666667 + 0.666667 i
-0.444444 + 1.19444 i
-0.606913 + 1.0646 i
-0.630766 + 1.09174 i
-0.629961 + 1.09112 i
-0.629961 + 1.09112 i
-0.629961 + 1.09112 i
-0.629961 + 1.09112 i
-0.629961 + 1.09112 i
-0.629961 + 1.09112 i
```

In[88]:= `NestList[NS, -I, 10] // N // Chop // TableForm`

Out[88]//TableForm=

```
-1. i
-0.666667 - 0.666667 i
-0.444444 - 1.19444 i
-0.606913 - 1.0646 i
-0.630766 - 1.09174 i
-0.629961 - 1.09112 i
-0.629961 - 1.09112 i
-0.629961 - 1.09112 i
-0.629961 - 1.09112 i
-0.629961 - 1.09112 i
-0.629961 - 1.09112 i
```

Három különbözô pontból indulva három kül. pontba tartanak a komplex iterációs sorozatok. Ezek éppen 3 (komplex) köbgyökei.

Konvergencia van 'majdnem' mindenhol, de hogy hová tartunk, nem nyilvánvaló: Ansatz, csak a 3 köbgyökhöz, ábrázoljuk egy medencét.
Az ábrázolás kényelemes a ListDensityPlot utasítással. Egy példa elôször: sakktábla

In[89]:=
```
ArrayPlot[Table[If[EvenQ[i + j], 1, 0], {j, -2, 2}, {i, -2, 2}]]
```

Out[89]=



In[90]:=
```
z /. Solve[z^3 == 2, z]
```

Out[90]= $\left\{-(-2)^{1/3},\ 2^{1/3},\ (-1)^{2/3}\,2^{1/3}\right\}$

In[91]:=
```
Arg[%] // N
```

Out[91]= {-2.0944, 0., 2.0944}

Argumetumok szerint színezünk. Fekete pontok: a valós köbgyök a limesz.

In[92]:=
```
Timing[ArrayPlot[Table[If[Abs[Arg[Nest[NS, i + j I, 6]] /. {Indeterminate → 10}] < 0.2, 1, 0],
   {j, -2., 2., GridSize}, {i, -2., 2., GridSize}], Mesh → False, ImageSize → {500, 500}]]
```

Out[92]= {0.375,  }

3 köbgyök medencéje 3 kül. színnel.

```
In[93]:= Timing[ArrayPlot[Table[θ = Arg[Nest[NS, i + j I, NoOfIterations] /. {Indeterminate → 10}];
    If[θ > 1.8, 1, If[θ < -1.8, 2, 0]], {j, -2., 2., GridSize},
    {i, -2., 2., GridSize}], Mesh → False, ImageSize → {500, 500}]]
```

Out[93]= {0.422,  }

```
In[94]:= MyCF[n_] := Which[n == 2, RGBColor[0, 0, 1], n == 0,
   RGBColor[1, 0, 0], n == 1, RGBColor[0, 1, 0], True, RGBColor[0, 0, 0]]
```

In[95]:=
```
Timing[ArrayPlot[Table[θ = Arg[Nest[NS, i + j I, NoOfIterations] /. {Indeterminate → 10}];
   If[θ > 1.8, 1, If[θ < -1.8, 2, 0]], {j, -2., 2., GridSize}, {i, -2., 2., GridSize}],
  Mesh → False, ImageSize → {500, 500}, ColorFunction → MyCF, ColorFunctionScaling → False]]
```

Out[95]= {0.406,  }

Vigyázat, csak saját felelôsségre [több mintapont, több iteráció]!

In[96]:=
```
NoOfIterations = 8;
```

In[97]:=
```
GridSize = .01;
```

In[98]:=
```
Timing[ArrayPlot[Table[θ = Arg[Nest[NS, i + j I, NoOfIterations] /. {Indeterminate → 10}];
   If[θ > 1.8, 1, If[θ < -1.8, 2, 0]], {j, -2., 2., GridSize}, {i, -2., 2., GridSize}],
  Mesh → False, ImageSize → {500, 500}, ColorFunction → MyCF, ColorFunctionScaling → False]]
```

Out[98]= {47.687, }

Megjegyzés Konv. biz. - , érdekesebb nemkonvergens "attraktorok" -

Színezés konvergencia gyorsasága szerint is

```
GridSize = .01;
```

```
T = Table[θ = NestWhileList[NS, i + j I, Unequal, 2, 40] /. {Indeterminate → 10};
   {Last[Arg[θ]], Length[θ]}, {j, -2., 2., GridSize}, {i, -2., 2., GridSize}];
```

```
MM = Max[Flatten[T, 1][[All, 2]]]
```

41

```
T2 = Partition[
   Map[(If[#[[1]] < -1.8, 0, If[#[[1]] > 1.8, 200, 100]] + #[[2]]) &, Flatten[T, 1]], Length[T]];
```

```
MyCF2[n_] := If[n ≥ 200, RGBColor[1 - Mod[n, 200] / MM, 0, 0],
  If[n ≥ 100, RGBColor[0, 1 - Mod[n, 100] / MM, 0], RGBColor[0, 0, 1 - n / MM]]]

Timing[ArrayPlot[T2, Mesh → False, ImageSize → {500, 500},
  ColorFunction → MyCF2, ColorFunctionScaling → False]]
```

{3.14,  }

Ez eddig empirikus matematika. De a konvergencia bizonyítása lehetséges fixpont-tételek felhasználásával (vonzó/taszító fixpontok, deriváltak)

Nemlineáris egyeneletrendszer megoldására is használható a NR-módszer (Jacobi mtx. inverze kell)

Newton Raphson (2D)

Hogyan módosul az alapfeladat és alapképlet?

$x - f[x] / f'[x] = x - f'[x]^{-1} f[x] \Longrightarrow X - J^{-1}[X] F[X]$

Példa

$x^2 - y^2 = 4$
$x^2 + y^2 = 16$

In[66]:= `f[x_, y_] := x^2 - y^2 - 4`

In[67]:= `g[x_, y_] := x^2 + y^2 - 16`

In[68]:= `F[{x_, y_}] := {f[x, y], g[x, y]}`

gradiensek

In[69]:= $\{\partial_x f[x, y], \partial_y f[x, y]\}$

Out[69]= {2 x, -2 y}

In[70]:= $\{\partial_x g[x, y], \partial_y g[x, y]\}$

Out[70]= {2 x, 2 y}

Jacobi

`Clear[J]`

In[71]:= $J[u\_, v\_] := \begin{pmatrix} \partial_x f[x, y] & \partial_y f[x, y] \\ \partial_x g[x, y] & \partial_y g[x, y] \end{pmatrix}$ /. {x → u, y → v}

In[72]:= `J[x, y] // MatrixForm`

Out[72]//MatrixForm=

$$\begin{pmatrix} 2\,x & -2\,y \\ 2\,x & 2\,y \end{pmatrix}$$

HF: Miért működik?

In[73]:= `Outer[D, {f[x, y], g[x, y]}, {x, y}]`

Out[73]= {{2 x, -2 y}, {2 x, 2 y}}

In[74]:= $\text{Inverse}\left[\begin{pmatrix} a & b \\ c & d \end{pmatrix}\right]$ // MatrixForm

Out[74]//MatrixForm=

$$\begin{pmatrix} \frac{d}{-b\,c+a\,d} & -\frac{b}{-b\,c+a\,d} \\ -\frac{c}{-b\,c+a\,d} & \frac{a}{-b\,c+a\,d} \end{pmatrix}$$

In[75]:= `Inverse[J[x, y]] // MatrixForm`

Out[75]//MatrixForm=

$$\begin{pmatrix} \frac{1}{4\,x} & \frac{1}{4\,x} \\ -\frac{1}{4\,y} & \frac{1}{4\,y} \end{pmatrix}$$

In[76]:= `X0 = {1, 2};`

In[77]:=
```
NR2Core[{x_, y_}] := {x, y} - Inverse[J[x, y]].F[{x, y}]
```

In[78]:=
```
NR2Core[{x, y}]
```

Out[78]= $\left\{x - \frac{-4 + x^2 - y^2}{4\,x} - \frac{-16 + x^2 + y^2}{4\,x},\ y + \frac{-4 + x^2 - y^2}{4\,y} - \frac{-16 + x^2 + y^2}{4\,y}\right\}$

In[79]:=
```
NestList[NR2Core, X0, 1]
```

Out[79]= $\left\{\{1, 2\}, \left\{\frac{11}{2}, \frac{5}{2}\right\}\right\}$

In[80]:=
```
NestList[NR2Core, X0, 6] // N // TableForm
```

Out[80]//TableForm=

1.	2.
5.5	2.5
3.65909	2.45
3.19601	2.44949
3.16246	2.44949
3.16228	2.44949
3.16228	2.44949

In[81]:=
```
Solve[{x^2 - y^2 == 4, x^2 + y^2 == 16}, {x, y}]
```

Out[81]= $\left\{\left\{x \to -\sqrt{10},\ y \to -\sqrt{6}\right\}, \left\{x \to -\sqrt{10},\ y \to \sqrt{6}\right\}, \left\{x \to \sqrt{10},\ y \to -\sqrt{6}\right\}, \left\{x \to \sqrt{10},\ y \to \sqrt{6}\right\}\right\}$

In[82]:=
```
N[%]
```

Out[82]= $\{\{x \to -3.16228,\ y \to -2.44949\}, \{x \to -3.16228,\ y \to 2.44949\},$
$\{x \to 3.16228,\ y \to -2.44949\}, \{x \to 3.16228,\ y \to 2.44949\}\}$

Az utolsó megoldáshoz konvergál az iteráltsorozat

Kaczmarz-Steinhaus

▪ 2D

KS-módszer lényege: tekintsük az Ax=b lin. egyenletrendszer egyenleteit geometriailag <ai,x>=bi ⟵ hipersíkok; metszéspontjuk koordinátái a megoldásvektor komponensei.
Ortogonális projekció minden egyenesre (2D), síkra (3D),...

A vetítéseket mátrix-vektor felírásban adjuk meg; ill. egy iterációban egymás után vetítünk az összes lehetséges módon P[X]=P2[-P1[X]], ahol P1 első egyenesre történő merőleges vet.
Áll: P1[x]=QQ1x+qq1 P2[x]=QQ2x+qq2, P[x]=QQx+qq

Képezzük a KS iterált-sorozatot, ha x0={5,0} és
2x+y=3
2x-y=2

?QQ1, QQ2, QQ ill. qq1, qq2, qq

ei oszlopvektor! a szokásos jelölések:
ai - A mtx. i-edik sora ei - hozzá tartozó normált vektor
bi- jobboldal i-edik komponense

QQi=I-ei ei*
qqi=bi/<ai,ai> ai

In[99]:=
```
A = (2 1; 2 -1); b = {3, 2}; x = {x1, x2};
```
$$A = \begin{pmatrix} 2 & 1 \\ 2 & -1 \end{pmatrix};\ b = \{3, 2\};\ x = \{x1, x2\};$$

In[100]:=
```
LinearSolve[A, b]
```

Out[100]= $\left\{\frac{5}{4}, \frac{1}{2}\right\}$

$$QQ1=I_2 - e1e1^* = \begin{pmatrix} 1 & 0 \\ 0 & 1 \end{pmatrix} - e1\,e1^* = \begin{pmatrix} 1 & 0 \\ 0 & 1 \end{pmatrix} - \begin{pmatrix} 4/5 & 2/5 \\ 2/5 & 1/5 \end{pmatrix} = \begin{pmatrix} 1/5 & -2/5 \\ -2/5 & 4/5 \end{pmatrix}$$

qq1=b1/<a1,a1>a1=3/5{2, 1}

In[101]:=
```
IdentityMatrix[2] -
  Transpose[{{2 / Sqrt[5], 1 / Sqrt[5]}}].{{2 / Sqrt[5], 1 / Sqrt[5]}} // MatrixForm
```

Out[101]//MatrixForm=
$$\begin{pmatrix} \frac{1}{5} & -\frac{2}{5} \\ -\frac{2}{5} & \frac{4}{5} \end{pmatrix}$$

$QQ2 = I_2 - e2e2^* = \begin{pmatrix} 1 & 0 \\ 0 & 1 \end{pmatrix} - e2\,e2^* = \begin{pmatrix} 1 & 0 \\ 0 & 1 \end{pmatrix} - \begin{pmatrix} 4/5 & -2/5 \\ -2/5 & 1/5 \end{pmatrix} = \begin{pmatrix} 1/5 & 2/5 \\ 2/5 & 4/5 \end{pmatrix}$

qq2=2/5 {2, −1}

In[102]:=
```
IdentityMatrix[2] -
  Transpose[{{2 / Sqrt[5], -1 / Sqrt[5]}}].{{2 / Sqrt[5], -1 / Sqrt[5]}} // MatrixForm
```

Out[102]//MatrixForm=

$$\begin{pmatrix} \frac{1}{5} & \frac{2}{5} \\ \frac{2}{5} & \frac{4}{5} \end{pmatrix}$$

In[103]:=

$QQ1 = \begin{pmatrix} 1/5 & -2/5 \\ -2/5 & 4/5 \end{pmatrix}$; $QQ2 = \begin{pmatrix} 1/5 & 2/5 \\ 2/5 & 4/5 \end{pmatrix}$; qq1 = 3 / 5 {2, 1}; qq2 = 2 / 5 {2, -1};

In[104]:= `QQ2.QQ1.x + QQ2.qq1 + qq2`

Out[104]= $\left\{\frac{32}{25} - \frac{3\,x1}{25} + \frac{6\,x2}{25}, \frac{14}{25} - \frac{6\,x1}{25} + \frac{12\,x2}{25}\right\}$

elsô és második vetítés kompozíciója (mátrix-vektor alak)

In[105]:= `QQ = QQ2.QQ1; qq = QQ2.qq1 + qq2;`

nulladik és elsô KS iterált az x0={5,0} kezdôpontból indulva

In[106]:= `NestList[(QQ.# + qq) &, {5, 0}, 1]`

Out[106]= $\left\{\{5, 0\}, \left\{\frac{17}{25}, -\frac{16}{25}\right\}\right\}$

In[107]:= `NestList[(QQ.# + qq) &, {5, 0}, 10] // N // TableForm`

Out[107]//TableForm=

5.	0.
0.68	−0.64
1.0448	0.0896
1.17613	0.352256
1.22341	0.446812
1.24043	0.480852
1.24655	0.493107
1.24876	0.497518
1.24955	0.499107
1.24984	0.499678
1.24994	0.499884

In[108]:= `LinearSolve[A, b]`

Out[108]= $\left\{\frac{5}{4}, \frac{1}{2}\right\}$

Megjegyzés. Csak az elsô egyenesre vetítés

In[109]:= `QQ1.x + qq1`

Out[109]= $\left\{\frac{6}{5}+\frac{x1}{5}-\frac{2\,x2}{5},\ \frac{3}{5}-\frac{2\,x1}{5}+\frac{4\,x2}{5}\right\}$

X={2,2} képe

In[110]:= `QQ1.{2, 2} + qq1`

Out[110]= $\left\{\frac{4}{5},\ \frac{7}{5}\right\}$

In[111]:=
```
Plot[3 - 2 x, {x, -1, 2}, AspectRatio → Automatic,
 Epilog → {PointSize[.03], Point[{2, 2}], RGBColor[1, 0, 0],
   Point[{4 / 5, 7 / 5}], RGBColor[0, 0, 0], Line[{{4 / 5, 7 / 5}, {2, 2}}]}]
```

Out[111]=





■ 3D

In[112]:=

$$A = \begin{pmatrix} 7 & 5 & 2 \\ 8 & 4 & 9 \\ 11 & 3 & 3 \end{pmatrix}; \; b = \{1, 2, 3\};$$

In[113]:= `LinearSolve[A, b]`

Out[113]= $\left\{\frac{39}{115}, -\frac{34}{115}, \frac{6}{115}\right\}$

In[114]:= `N[%]`

Out[114]= {0.33913, -0.295652, 0.0521739}

A sorainak normái

In[115]:= `Sqrt[Plus @@ A[[1]]^2] // N`

Out[115]= 8.83176

In[116]:= $\text{MyNorm}[v\_] := \sqrt{\sum_{i=1}^{\text{Length}[v]} v[[i]]^2}$

In[117]:= `MyNorm[A[[1]]]`

Out[117]= $\sqrt{78}$

```
In[118]:= N[%]

Out[118]= 8.83176

In[119]:= norma1 = N[MyNorm[A[[1]]]];
          norma2 = N[MyNorm[A[[2]]]];
          norma3 = N[MyNorm[A[[3]]]];

In[122]:= norma3

Out[122]= 11.7898

In[123]:= 1 / norma1 A[[1]]

Out[123]= {0.792594, 0.566139, 0.226455}

In[124]:= 1 / norma2 A[[2]]

Out[124]= {0.630488, 0.315244, 0.709299}

In[125]:= 1 / norma3 A[[3]]

Out[125]= {0.933008, 0.254457, 0.254457}

In[126]:= e1 = A[[1]] / norma1; e2 = A[[2]] / norma2; e3 = A[[3]] / norma3;

In[127]:= β1 = b[[1]] / norma1; β2 = b[[2]] / norma2; β3 = b[[3]] / norma3;

In[128]:= e1

Out[128]= {0.792594, 0.566139, 0.226455}

In[129]:= Norm[e1]

Out[129]= 1.

In[130]:= QQ1 = IdentityMatrix[3] - (Transpose[{e1}].{e1}); qq1 = β1 e1;

In[131]:= QQ2 = IdentityMatrix[3] - (Transpose[{e2}].{e2}); qq2 = β2 e2;

In[132]:= QQ3 = IdentityMatrix[3] - (Transpose[{e3}].{e3}); qq3 = β3 e3;

In[133]:= qq = QQ3.QQ2.qq1 + QQ3.qq2 + qq3;

In[134]:= QCore[x_] := QQ3.QQ2.QQ1.x + qq;

In[135]:= x = {0, 0, 0};

In[136]:= NestList[QCore, x, 100] // TableForm

Out[136]//TableForm=
0         0           0
0.217603  0.10795     0.0941737
0.223509  0.0654835   0.114984
0.230971  0.026077    0.127029
0.23925   -0.0100844  0.132833
0.247826  -0.042975   0.134278
0.256341  -0.0726768  0.132759
0.264557  -0.099341   0.129298
0.272324  -0.123161   0.124642
0.279553  -0.144354   0.119325
0.286205  -0.163143   0.113726
```

0.292266	-0.179752	0.108108
0.297749	-0.194396	0.102648
0.302678	-0.207279	0.0974586
0.307086	-0.218592	0.0926082
0.311012	-0.22851	0.088132
0.314495	-0.237192	0.0840422
0.317576	-0.244782	0.0803353
0.320295	-0.251411	0.0769974
0.322687	-0.257195	0.0740079
0.324789	-0.262237	0.0713426
0.326633	-0.266628	0.0689754
0.328247	-0.270452	0.0668796
0.329659	-0.273778	0.0650293
0.330892	-0.27667	0.0633995
0.331968	-0.279183	0.0619671
0.332906	-0.281367	0.0607102
0.333724	-0.283264	0.0596091
0.334436	-0.28491	0.0586458
0.335055	-0.28634	0.057804
0.335594	-0.28758	0.0570692
0.336062	-0.288655	0.0564284
0.336469	-0.289588	0.05587
0.336822	-0.290397	0.0553837
0.337129	-0.291099	0.0549605
0.337395	-0.291707	0.0545924
0.337626	-0.292234	0.0542724
0.337826	-0.292691	0.0539943
0.338	-0.293087	0.0537528
0.338151	-0.29343	0.053543
0.338282	-0.293728	0.053361
0.338395	-0.293985	0.053203
0.338493	-0.294208	0.0530659
0.338579	-0.294402	0.052947
0.338652	-0.294569	0.0528439
0.338716	-0.294714	0.0527545
0.338772	-0.29484	0.0526769
0.33882	-0.294949	0.0526097
0.338861	-0.295043	0.0525515
0.338897	-0.295125	0.052501
0.338929	-0.295195	0.0524573
0.338956	-0.295257	0.0524193
0.338979	-0.29531	0.0523865
0.338999	-0.295355	0.052358
0.339017	-0.295395	0.0523334
0.339032	-0.29543	0.052312
0.339045	-0.29546	0.0522935
0.339057	-0.295485	0.0522775
0.339067	-0.295508	0.0522636
0.339075	-0.295527	0.0522516
0.339083	-0.295544	0.0522412
0.339089	-0.295558	0.0522322
0.339095	-0.295571	0.0522244

0.339099	-0.295582	0.0522176
0.339104	-0.295591	0.0522118
0.339107	-0.295599	0.0522067
0.33911	-0.295607	0.0522023
0.339113	-0.295613	0.0521985
0.339115	-0.295618	0.0521952
0.339117	-0.295623	0.0521923
0.339119	-0.295627	0.0521899
0.339121	-0.29563	0.0521877
0.339122	-0.295633	0.0521859
0.339123	-0.295636	0.0521843
0.339124	-0.295638	0.0521829
0.339125	-0.29564	0.0521817
0.339126	-0.295641	0.0521806
0.339126	-0.295643	0.0521797
0.339127	-0.295644	0.052179
0.339127	-0.295645	0.0521783
0.339128	-0.295646	0.0521777
0.339128	-0.295647	0.0521772
0.339128	-0.295648	0.0521767
0.339129	-0.295648	0.0521764
0.339129	-0.295649	0.052176
0.339129	-0.295649	0.0521758
0.339129	-0.29565	0.0521755
0.339129	-0.29565	0.0521753
0.33913	-0.29565	0.0521751
0.33913	-0.295651	0.0521749
0.33913	-0.295651	0.0521748
0.33913	-0.295651	0.0521747
0.33913	-0.295651	0.0521746
0.33913	-0.295651	0.0521745
0.33913	-0.295651	0.0521744
0.33913	-0.295651	0.0521743
0.33913	-0.295652	0.0521743
0.33913	-0.295652	0.0521742
0.33913	-0.295652	0.0521742
0.33913	-0.295652	0.0521742
0.33913	-0.295652	0.0521741

Feladat

Feladat
a) síkbeli forgás illusztrációja
b) interaktív szines viz síkbeli vetületeknek.
c) 3D

In[137]:=
$$TM[\alpha\_] := \begin{pmatrix} Cos[\alpha] & -Sin[\alpha] \\ Sin[\alpha] & Cos[\alpha] \end{pmatrix}$$

In[138]:=
```
Manipulate[Show[Graphics[{Black, PointSize[.02], Point[P]}],
  Graphics[{Red, PointSize[.02], Point[TM[α].P]}],
  Graphics[{Blue, Line[{P, {0, 0}, TM[α].P}]}], PlotRange → {{-2, 2}, {-2, 2}}, Axes -> True],
 {P, {-1, -1}, {1, 1}, Slider2D}, {{α, Pi / 2}, 0, Pi, .2}]
```

Out[138]=



In[139]:=
```
Clear[A, b];
```
$A = \begin{pmatrix} 2 & 1 \\ 2 & -1 \end{pmatrix}$; b = {3, 2};

projekciós mátrixokat és vektorokat előállító függvények

In[141]:=
```
PM[M_, n_] :=
  Module[{ei = M[[n]] / Norm[M[[n]]]}, IdentityMatrix[Length[M]] - Transpose[{ei}].{ei}];
PV[M_, v_, n_] := v[[n]] / (M[[n]].M[[n]]) M[[n]];
```

In[143]:= `PM[A, 1] // MatrixForm`

Out[143]//MatrixForm=

$$\begin{pmatrix} \frac{1}{5} & -\frac{2}{5} \\ -\frac{2}{5} & \frac{4}{5} \end{pmatrix}$$

In[144]:= `PV[A, b, 1]`

Out[144]= $\left\{\frac{6}{5}, \frac{3}{5}\right\}$

In[145]:= `PM[A, 2]`

Out[145]= $\left\{\left\{\frac{1}{5}, \frac{2}{5}\right\}, \left\{\frac{2}{5}, \frac{4}{5}\right\}\right\}$

In[146]:= `PV[A, b, 2]`

Out[146]= $\left\{\frac{4}{5}, -\frac{2}{5}\right\}$

Interaktív játék

Plot + Epilog: mozgatható pont (Slider2D), vetületi pontok, összekötő szakaszok

In[151]:=
```
Manipulate[Plot[{(b[[1]] - A[[1, 1]] x) / A[[1, 2]], (b[[2]] - A[[2, 1]] x) / A[[2, 2]]},
  {x, -3, 3}, PlotRange → {{-3, 3}, {-3, 3}}, AspectRatio → Automatic,
  Epilog → {PointSize[.02], Point[P], Red, Point[PM[A, 1].P + PV[A, b, 1]], Blue,
    Point[PM[A, 2].P + PV[A, b, 2]], Black, Line[{P, PM[A, 1].P + PV[A, b, 1]}],
    Line[{P, PM[A, 2].P + PV[A, b, 2]}]}], {P, {-2, -2}, {2, 2}, Slider2D}]
```

Out[151]=